UR1981-34

1981/34. FORTRAN programs for performing gravity reduction

R.G. Richardson

Abstract

The main program calculates observed Bouguer anomalies from station position, gravity difference, and terrain correction data. Options are available for most forms of data specification. A short program performs conversions between the kiloyard and metric grids.

INTRODUCTION

The programs were adapted for use on the Department of Mines Perkin-Elmer 6/16 computer from ALGOL versions running on the ELLIOTT 503 and Burroughs B6700 computers at the University of Tasmania. The FORTRAN versions described here may be easily transferred to other computers.

THE PROGRAMS

U191V6 (Appendix 1)

This program implements the gravity reduction method outlined in Leaman (1981) and uses the old equation for normal gravity. By selecting the appropriate control data, station positions may be in metric grid co-ordinates or latitude and longitude, heights may be in feet or metres, and the observed gravity may be in milligals or meter scale divisions. The data is terminated by a station number less than -9000.0. The use of the integer variable FFEED is machine dependent.

Input data is:

PA1, PA3, PA4, PA5 - format (I4, F12.4, F10.2, A8)
- PA1 is the survey number and has the form (for the Department of Mines) XX51 where XX is the year of the survey
- PA3 is the number of the reference station
- PA4 is the absolute gravity value at the reference station (milligals)
- PA5 is the date of reference station establishment *e.g.* 041181

PA6, SCONST, PA8 - format (A4, F10.5, A8)
- PA6 is the meter number *e.g.* W273, S308
- SCONST is the meter scale constant in milligals per scale division
- PA8 is the meter calibration date *e.g.* 030180

PA9, PA12, PA14, PA16, PA17 - format (5I1)
- PA9 = 1 - input elevations in metres
  = 0 - input elevations in feet
- PA12 = 1 - input units milligals
  = 0 - input units scale divisions
- PA14 = 1 - input in metric grid co-ordinates
  = 0 - input in latitude and longitude
- PA16 = 1 - output elevations in metres
  = 0 - output elevations in feet
- PA17 = 1 - output in grid co-ordinates
  = 0 - output in latitude and longitude

RHO, HTCOR
- format (F6.2, F9.2)
- RHO is the Bouguer density (t/m^3)
- HTCOR is any elevation datum change that might be required

TITLE
- format (18A4)
- the survey title

For each station:

(i) input in grid co-ordinates:

A1, EAST, NORTH, ELEV, GOBS, ABC, CORR - format (F9.4, 2F10.3, F8.2, F10.2, 2A4, A2, F7.2)
- A1 is the station number in the form 8051.1234 or .1234
- EAST is the east co-ordinate of the station (in metres)
- NORTH is the north co-ordinate of the station (in metres)
- ELEV is the station elevation
- GOBS is the observed gravity difference
- ABC is an alphanumeric string
- CORR is any correction - *e.g.* terrain correction

(ii) input in latitude and longitude:

A1, LATD, LATM, LATS, LOND, LONM, LONS, ELEV, GOBS, ABC, CORR
- format (F9.4, I3, 2F6.0, I4, 2F6.0, F8.2, F10.2, 2A4, A2, F7.2)
- A1 is the station number in the form 8051.1234 or .1234
- LATD is the latitude degrees (integer)
- LATM is the latitude minutes (real or integer)
- LATS is the latitude seconds (may be zero if LATM is real)
- LOND is the longitude degrees (integer)
- LONM is the longitude minutes (real or integer)
- LONS is the longitude seconds (may be zero if LONM is real)
- ELEV is the station elevation
- GOBS is the observed gravity difference
- ABC is an alphanumeric string
- CORR is any correction - *e.g.* terrain correction

After the last station of a data set -99999. should appear instead of the station number.

Appendices 2, 3 and 4 show sample inputs and outputs.

KYD2KM (Appendix 5)

This program converts co-ordinates from the kiloyard grid to the metric grid. Data is prepared in the format used for U191V6 with co-ordinates in yards instead of metres. The program converts the co-ordinates and produces an output suitable for U191V6.

3/21

REFERENCE

LEAMAN, D.E. 1981. Gravity survey and reduction techniques. *Geophys. Spec.Rep.Dep.Mines Tasm.* 9.

[17 June 1981]

APPENDIX 1

Program U191V6


```
C USED TO CALCULATE GRAVITY REDUCTIONS GIVEN ELEVATIONS,
C COORDS(LAT & LONG OR METRIC GRID), BASE VALUES, METER READINGS
C OR OBSERVED GRAVITIES AND TERRAIN CORRECTIONS.
C RICHARDSON, 1981
C
      REAL PA6
      INTEGER*2 PA1, PA17, PA9, PA10, PA12, PA13, PA14, PA15, NA,
     . PA16, NL, PA91, PA141, PA121, PA161, PA171
      DOUBLE PRECISION PA4, SCONST, PA5, PA3, PA8, RHO, HTCOR
      REAL TITLE(18)
      DOUBLE PRECISION A1, A2GRID, A5GRID, A2, A3, A4, A5, A6, A7, CORR, A2T,
     . A3T, A4T, ELEV, GOBS, GTHEO, FAA, BA, GBASE
      INTEGER*2 N, IA2, IA3, IA5, IA6
      REAL ABC(3), AR(6)
      INTEGER*2 FFEED
      DATA FFEED/3072/
C USE FOR FORM FEED
C
C THE PROGRAM BODY
      READ(4, 100) PA1, PA3, PA4, PA5
  100 FORMAT(I4, F12.4, F10.2, A8)
C   PA1=SURVEY NUMBER
C   PA3=REFERENCE STATION NUMBER
C   PA4=REFERENCE STATION VALUE(MGALS)
C   PA5=DECIMAL DATE OF BASE STATION ESTABLISHMENT
      READ(4, 101) PA6, SCONST, PA8
  101 FORMAT(A4, F10.5, A8)
C   PA6=METER NUMBER
C   SCONST=SCALE CONSTANT
C   PA8=METER CALIBRATION DATE
      READ(4, 102) PA9, PA12, PA14, PA16, PA17
  102 FORMAT(5I1)
      PA10=1-PA9
      PA13=1-PA12
      IF (PA14 .EQ. 0) PA17=0
C IF INPUT IN LAT AND LONG CAN'T OUTPUT GRID COORDS.
C   PA9  =1 INPUT ELEVATION METRES
C        =0 INPUT ELEVATION FEET
C   PA12 =1 INPUT MGALS
C        =0 INPUT SCALE DIVISIONS RELATIVE TO BASE STATION
C   PA14 =1 INPUT GRID COORDS
C        =0 INPUT LAT/LONG
C   PA16 =1 OUTPUT IN METRES
C        =0 OUTPUT IN FEET
C   PA17 =1 OUTPUT IN GRID COORDS
C        =0 OUTPUT IN LAT/LONG
      READ(4, 104) RHO, HTCOR
  104 FORMAT(F6.2, F9.2)
C   RHO = BOUGUER DENSITY(T/M*M*M)
C   HTCOR = AN ELEVATION DATUM CHANGE
      READ(4, 105) TITLE
  105 FORMAT(18A4)
```

```
      WRITE(6,200) FFEED
  200 FORMAT(A1)
C WRITE A FORMFEED TO THROW A NEW PAGE
C ADD SIMPLE VARIABLES TO COPE WITH 1-PA9 ETC.  IN PRINT STATEMENT
      PA91=1-PA9
      PA141=1-PA14
      PA121=1-PA12
      PA161=1-PA16
      PA171=1-PA17
      WRITE(6,201) PA1,PA3,PA4,PA5,PA6,SCONST,PA8,PA9,PA91,
     . PA14,PA141,PA12,PA121,RHO,PA16,PA161,HTCOR,
     . PA17,PA171,TITLE
  201 FORMAT(30X,'SURVEY'/34X,'NUMBER',15X,I7//30X,'DATUM'/
     . 34X,'REFERENCE STATION',4X,F12.4/34X,'VALUE (IN MGALS)',
     . 7X,F10.2/34X,'DATE (DECIMAL)',8X,A8//30X,'METER'/34X,
     . 'NUMBER',18X,A4/34X,'SCALE CONSTANT',5X,F12.5/34X,
     . 'CALIBRATION DATE',6X,A8//30X,'FIELD DATA'/34X,'ELEVATION'/
     . 38X,'IN METRES',9X,I3/38X,'IN FEET',11X,I3/34X,'STATION',
     . ' POSITIONS'/38X,'IN GRID COORDS',4X,I3/38X,'IN LAT/LONG',
     . 7X,I3/34X,'OBSERVED GRAVITY'/38X,'IN MGALS',10X,I3/38X,
     . 'IN SCALE DIVS',5X,I3//30X,'OUTPUT DATA'/34X,'BOUGUER',
     . ' DENSITY',5X,F8.2/
     . 34X,'ELEVATION'/38X,'IN METRES',9X,I3/38X,'IN FEET',11X,
     . I3/34X,'ELEVATION DATUM CHANGE',F12.2/34X,'STATION ',
     . 'POSITIONS'/38X,'IN GRID COORDS',4X,I3/38X,'IN LAT/LONG',
     . 7X,I3///20X,18A4)
      N=0
      PA15=0
      GBASE=PA4*1.0D-03
      CALL HEADNG(PA17)
   51 CONTINUE
      IF (PA14 .EQ. 1) READ(4,106) A1,A2GRID,A5GRID,ELEV,GOBS,ABC,CORR
  106 FORMAT(F9.4,2F10.3,F8.2,F10.2,2A4,A2,F7.2)
C     A1 = STATION NUMBER
C     A2GRID = EAST GRID COORD
C     A5GRID = NORTH GRID COORD
C     ELEV = ELEVATION
C     GOBS = OBSERVED GRAVITY
C     ABC = TITLE OF LESS THAN 10 CHARACTERS
C     CORR = CORRECTION TO OBSERVED GRAVITY
      IF (PA14 .EQ. 0) READ(4,107) A1,A2T,A3T,A4T,A5,A6,A7,ELEV,
     . GOBS,ABC,CORR
  107 FORMAT(F9.4,I3,2F6.0,I4,2F6.0,F8.2,F10.2,2A4,A2,F7.2)
C     A1 = STATION NUMBER
C     A2 = LATDEG
C     A3 = LAT MIN - MAY BE REAL
C     A4 = LAT SEC
C     A5 = LONG DEG
C     A6 = LONG MIN -  MAY BE REAL
C     A7 = LONG SEC
C     ELEV = ELEVATION
C     GOBS = OBSERVED GRAVITY
```

```
C     ABC = TITLE
C     CORR = CORRECTION TO OBSERVED GRAVITY
      IF (A1 .LT. -9000.0D00) GOTO 71
      IF (A1 .LT. 1.0D00) A1=A1+PA1
C IF THE STATION NUMBER IS LESS THAN NUMBER THEN ADD 1 TO THE STATION
C NUMBER
C IF THE STATION NUMBER IS LESS THAN 1 ADD THE SURVEY NUMBER
      PA15=PA15+1
C INCREMENT THE COUNTER FO THE NUMBER OF STATIONS
      GOBS=GOBS*(DBLE(FLOAT(PA12))+DBLE(FLOAT(PA13))*SCONST)
C CONVERT FROM SCALE DIVISIONS TO MGALS
      ELEV=ELEV+HTCOR
C APPLY CHANGE IN ELEVATION DATUM
      ELEV=ELEV*(DBLE(FLOAT(PA9))+0.3047997D00*DBLE(FLOAT(PA10)))
C CONVERT FEET TO METRES
      IF (PA14 .EQ. 0) GOTO 10
C IF PA14=1 HAVE INPUT IN GRID COORDS
      CALL GEOGRF(A2GRID,A5GRID,AR)
      A2T=AR(1)
      A3T=AR(2)
      A4T=AR(3)
      A5=AR(4)
      A6=AR(5)
      A7=AR(6)
      GOTO 11
C
C IF IN DEGREES CONVERT DECIMAL MINUTES TO MINUTES AND SECONDS
   10 CONTINUE
      CALL CONVRT(A3T,A4T)
      CALL CONVRT(A6,A7)
   11 CONTINUE
      A2=A2T
      A3=A3T
      A4=A4T
      GOBS=(GOBS+CORR)*1.0D-03+GBASE
      A2=(A2+A3/60.0D00+A4/3600.0D00)*0.0174532925D+00
      A3=DSIN(A2)
      GTHEO=978.0490D00+5.172314332D00*A3*A3-0.0057704891D00*
     . 4.0D00*A3*A3*(1.0D00-A3*A3)
      FAA=0.000308772D00*ELEV*(1.0D00-0.001422D00*A3*A3)
     .     +7.2D-11*ELEV*ELEV+GOBS-GTHEO
      BA=-6.2831853D00*RHO*6.67D-06*ELEV
      BA=FAA+BA
      IF (PA16 .EQ. 0) ELEV=ELEV*3.2808399D00
C ELEVATION BACK TO FEET IF NEEDED
      GOBS=GOBS-CORR*1.0D-03
      FAA=FAA-CORR*1.0D-03
      BA=BA*1.0D03
      IA2=A2T
      IA3=A3T
      IA5=A5
      IA6=A6
```

```
      IF (PA17 .EQ. 1) WRITE(6,202) A1,A2GRID,A5GRID,ELEV,GOBS,
     .  GTHEO,CORR,BA,ABC
  202 FORMAT(F12.4,2F10.0,F10.2,2F11.5,2F8.2,4X,2A4,A2)
      IF (PA17 .EQ. 0) WRITE(6,203) A1,IA2,IA3,A4T,IA5,IA6,A7,
     .  ELEV,GOBS,GTHEO,CORR,BA,ABC
  203 FORMAT(F12.4,2I4,F6.1,I5,I4,F6.1,F10.2,2F11.5,2F8.2,4X,2A4,A2)
      N=N+1
      IF (N .LT. 55) GOTO 71
      CALL HEADNG(PA17)
      N=0
   71 CONTINUE
      IF (A1 .GT. 0.0) GOTO 51
      WRITE(6,204) PA15
  204 FORMAT(/I5)
      STOP
      END
```



```
      SUBROUTINE DECDEG(ANGLE,IDEG,MIN,SEC)
      DOUBLE PRECISION ANGLE,SEC
      INTEGER*2 IDEG,MIN
      DOUBLE PRECISION DEGI,RMIN
      IDEG=IFIX(SNGL(ANGLE))
      DEGI=IDEG
      ANGLE=60.0D00*(ANGLE-DEGI)
      MIN=IFIX(SNGL(ANGLE))
      RMIN=MIN
      SEC=60.0D00*(ANGLE-RMIN)
      RETURN
      END
DECDEG 0024R  FUNC/SUB
DECDEG 0122R  FUNC VAR
.Q     0000R  EXT FUNC
.P     0000R  EXT FUNC
ANGLE  002AR  FORM PAR
IDEG   002CR  FORM PAR
MIN    002ER  FORM PAR
SEC    0030R  FORM PAR
DEGI   012AR  DBLE VAR
RMIN   0132R  DBLE VAR
IFIX   0000R  EXT FUNC
SNGL   0000R  EXT FUNC
.X     0000R  EXT FUNC
@P     0000R  EXT FUNC
@K     0000R  EXT FUNC
@L     0000R  EXT FUNC
@S     0000R  EXT FUNC
@M     0000R  EXT FUNC

0000 ERRORS
```

```
      SUBROUTINE GEOGRF(EAST,NORTH,ANSWER)
      DOUBLE PRECISION NORTH,EAST
      REAL ANSWER(6)
      DOUBLE PRECISION TMD1,ORTH,DEG,TMD2,CON,Q1,DOG,SIND,COSD,
     . TAND,TANSQ,SINSEC,C,D,DTAN,R,P,PCUBE,POR,PFIVE,P1,P2,P3,P4,
     . P5,GLAT,TEMPL,DLONG,DC,SLO,SLA
      DOUBLE PRECISION VAL(16),Q(5)
      INTEGER*2 I,IDLA,MLA,IDLO,MLO
C CONVERTS AMG GRID COORDS (IN METRES) TO LATS AND LONGS
C INPUT OF NORTH COORDINATE EXCLUDES LEADING 5 FOR USE IN TASMANIA
C E.G.  5326446.014 BECOMES 326446.014
      VAL(1)=9.033D-06
      VAL(2)=1.1113334878D05
      VAL(3)=3.4906585D-02
      VAL(4)=6.981317D-02
      VAL(5)=1.04719755D-01
      VAL(6)=9.008D-06
      VAL(7)=1.745329252D-02
      VAL(8)=0.66945418546D-02
      VAL(9)=1.99840032D-08
      VAL(10)=23.96162303D-20
      VAL(11)=5.9928028794D-14
      VAL(12)=119.76019192D-26
      VAL(13)=2.9964014397D-14
      VAL(14)=14.970023989D-26
      VAL(15)=6.3354611409D06
      VAL(16)=1.60389546D04
C
      ORTH=NORTH+5.0D06
C ADD THE LEADING 5 TO THE NORTH COORDINATE
      TMD1=1.0D07-ORTH
      DEG=VAL(1)*TMD1
C
      DO 10 I=1,10
      TMD2=VAL(2)*DEG-VAL(16)*DSIN(VAL(3)*DEG)+
     . 16.83310D00*DSIN(VAL(4)*DEG)-0.0218D00*DSIN(VAL(5)*DEG)
      CON=TMD1-0.9996D00*TMD2
      IF (DABS(CON) .LT. 1.0D-03) GOTO 14
      DEG=DEG+VAL(6)*CON
   10 CONTINUE
      GOTO 16
C
   14 CONTINUE
      Q1=(EAST-500000.0D00)*1.0D-06
      Q(1)=Q1
      DO 11 I=2,5
      Q(I)=Q(I-1)*Q1
   11 CONTINUE
      DOG=VAL(7)*DEG
      SIND=DSIN(DOG)
      COSD=DCOS(DOG)
      TAND=SIND/COSD
```

```
      TANSQ=TAND*TAND
      SINSEC=0.48481368111D-05
      C=1.0D00-VAL(8)*SIND*SIND
      D=0.2777777778D01
      DTAN=D*TAND
      R=VAL(15)/C**1.5
      P=637816.0D01/C**0.5
      PCUBE=P**3
      POR=P/R
      PFIVE=PCUBE*P*P
      P1=DTAN/(VAL(9)*R*P*SINSEC)
      P2=DTAN*(5.0D00+3.0D00*TANSQ)/(VAL(10)*R*PCUBE*SINSEC)
      P3=D/(P*COSD*SINSEC*.9996D-02)
      P4=D*(POR+2.0D00*TANSQ)/(VAL(11)*PCUBE*SINSEC*COSD)
      P5=D*(5.0D00+TANSQ*(28.0D00+24.0D00*TANSQ))/(VAL(12)*
     *    COSD*PFIVE*SINSEC)
      GLAT=DEG-P1*Q(2)+P2*Q(4)
      TEMPL=P3*Q1-P4*Q(3)+P5*Q(5)
      DLONG=TEMPL+7.0D00
      CALL DECDEG(DLONG,IDLO,MLO,SLO)
      IDLO=IDLO+140
      CALL DECDEG(GLAT,IDLA,MLA,SLA)
      GOTO 18
   16 CONTINUE
      WRITE(5,100)
  100 FORMAT(' ********** GEOGRAPHICAL ERROR **********')
      STOP 9998
   18 CONTINUE
      ANSWER(1)=IDLA
      ANSWER(2)=MLA
      ANSWER(3)=SLA
      ANSWER(4)=IDLO
      ANSWER(5)=MLO
      ANSWER(6)=SLO
C STORE LAT AND LONG READY TO RETURN
      RETURN
      END
```

34-A1/7

```
      SUBROUTINE CONVRT(A,B)
      DOUBLE PRECISION A,B
      INTEGER*2 I
      A=A+(B/60.0D00)
      I=IFIX(SNGL(A))
      B=(A-DBLE(FLOAT(I)))*60.0D00
      A=I
      RETURN
      END
CONVRT 0024R   FUNC/SUB
CONVRT 00E8R   FUNC VAR
.0     0000R   EXT FUNC
.P     0000R   EXT FUNC
A      002AR   FORM PAR
B      002CR   FORM PAR
I      00F2R   INT2 VAR
@K     0000R   EXT FUNC
@L     0000R   EXT FUNC
@D     0000R   EXT FUNC
@A     0000R   EXT FUNC
@P     0000R   EXT FUNC
IFIX   0000R   EXT FUNC
SNGL   0000R   EXT FUNC
DBLE   0000R   EXT FUNC
FLOAT  0000R   EXT FUNC
@1     0000R   EXT FUNC
@M     0000R   EXT FUNC
.X     0000R   EXT FUNC

0000 ERRORS
```

```
      SUBROUTINE HEADNG(PA17)
      INTEGER*2 PA17,FFEED
      DATA FFEED/3072/
      WRITE(6,100) FFEED
  100 FORMAT(A1)
      IF (PA17 .EQ. 0) WRITE(6,102)
  102 FORMAT('   ST NUMBER    LATITUDE       LONGITUDE',
     . 6X,'HEIGHT  OBS GRAV   THEO GRAV   CORRN',4X,'BA',7X,
     . 'COMMENT'/)
      IF (PA17 .EQ. 1) WRITE(6,113)
  113 FORMAT(3X,'ST NUMBER      EAST',5X,'NORTH',5X,
     . 'HEIGHT  OBS GRAV   THEO GRAV',3X,'CORRN',4X,'BA',
     . 7X,'COMMENT'/)
      RETURN
      END
HEADNG 0024R  FUNC/SUB
HEADNG 019AR  FUNC VAR
.Q     0000R  EXT FUNC
.P     0000R  EXT FUNC
PA17   002AR  FORM PAR
FFEED  0030R  INT2 VAR
100    0052R  LABEL
@H     0000R  EXT FUNC
102    0084R  LABEL
113    011ER  LABEL

0000 ERRORS
```

APPENDIX 2

Sample data, latitude input

```
8051    8051.9900 980169.78  181280
W913     0.0940  131278
10010
  2.67
QUE RIVER MINE AREA 1980
    .6001 41    35  57.4 145    41  47.5   695.4     10.53            1.47
    .6002 41 35.93       145 41.73          683.7     28.08            1.48
-99999.
```

```
SURVEY
    NUMBER                      8051

DATUM
    REFERENCE STATION           8051.9900
    VALUE (IN MGALS)            980169.78
    DATE (DECIMAL)              181280

METER
    NUMBER                      W913
    SCALE CONSTANT              0.09400
    CALIBRATION DATE            131278

FIELD DATA
    ELEVATION
        IN METRES               1
        IN FEET                 0
    STATION POSITIONS
        IN GRID COORDS          0
        IN LAT/LONG             1
    OBSERVED GRAVITY
        IN MGALS                0
        IN SCALE DIVS           1

OUTPUT DATA
    BOUGUER DENSITY             2.67
    ELEVATION
        IN METRES               1
        IN FEET                 0
    ELEVATION DATUM CHANGE              0.00
    STATION POSITIONS
        IN GRID COORDS          0
        IN LAT/LONG             1

QUE RIVER MINE AREA 1980
```

ST NUMBER	LATITUDE			LONGITUDE			HEIGHT	OBS GRAV	THEO GRAV	CORRN	BA	COMMENT
8051.6001	41	35	57.4	145	41	47.5	695.40	980.17077	980.32319	1.47	-14.14	
8051.6002	41	35	55.8	145	41	43.8	683.70	980.17242	980.32315	1.48	-14.75	

2

APPENDIX 3

Sample data, grid co-ordinate output

34-A3/1

```
8051    8051.9900 980169.78  181280
W913     0.0940  131278
10111
  2.67  0.0
QUE RIVER MINE AREA GRAVITY SURVEY     DECEMBER 1980
    .6001    391373.    393878.    695.4      10.53              1.47
    .6002    391283.    393926.    683.7      28.08              1.48
    .6003    391396.    393869.    698.5       7.10              1.44
    .6004    391419.    393860.    699.2       3.32              1.44
    .6005    391442.    393850.    700.6      -0.96              2.00
-99999.
```

```
SURVEY
    NUMBER                      8051

DATUM
    REFERENCE STATION           8051.9900
    VALUE (IN MGALS)            980169.78
    DATE (DECIMAL)              181280

METER
    NUMBER                      W913
    SCALE CONSTANT              0.09400
    CALIBRATION DATE            131278

FIELD DATA
    ELEVATION
        IN METRES               1
        IN FEET                 0
    STATION POSITIONS
        IN GRID COORDS          1
        IN LAT/LONG             0
    OBSERVED GRAVITY
        IN MGALS                0
        IN SCALE DIVS           1

OUTPUT DATA
    BOUGUER DENSITY             2.67
    ELEVATION
        IN METRES               1
        IN FEET                 0
    ELEVATION DATUM CHANGE          0.00
    STATION POSITIONS
        IN GRID COORDS          1
        IN LAT/LONG             0
```

ST NUMBER	EAST	NORTH	HEIGHT	OBS GRAV	THEO GRAV	CORRN	BA	COMMENT
8051.6001	391373.	393878.	695.40	980.17077	980.32319	1.47	-14.14	
8051.6002	391283.	393926.	683.70	980.17242	980.32315	1.48	-14.75	
8051.6003	391396.	393869.	698.50	980.17045	980.32320	1.44	-13.89	
8051.6004	391419.	393860.	699.20	980.17009	980.32321	1.44	-14.12	
8051.6005	391442.	393850.	700.60	980.16969	980.32321	2.00	-13.69	

5

34-A3/2

16/21

APPENDIX 4

Sample data, latitude output

```
8051    8051.9900 980169.78  181280
W913     0.0940  131278
10110
  2.67  0.0
QUE RIVER MINE AREA GRAVITY SURVEY     DECEMBER 1980
    .6001    391373.    393878.    695.4      10.53            1.47
    .6002    391283.    393926.    683.7      28.08            1.48
    .6003    391396.    393869.    698.5       7.10            1.44
    .6004    391419.    393860.    699.2       3.32            1.44
    .6005    391442.    393850.    700.6      -0.96            2.00
-99999.
```

SURVEY	
NUMBER	8051
DATUM	
REFERENCE STATION	8051.9900
VALUE (IN MGALS)	980169.78
DATE (DECIMAL)	181280
METER	
NUMBER	W913
SCALE CONSTANT	0.09400
CALIBRATION DATE	131278
FIELD DATA	
ELEVATION	
IN METRES	1
IN FEET	0
STATION POSITIONS	
IN GRID COORDS	1
IN LAT/LONG	0
OBSERVED GRAVITY	
IN MGALS	0
IN SCALE DIVS	1
OUTPUT DATA	
BOUGUER DENSITY	2.67
ELEVATION	
IN METRES	1
IN FEET	0
ELEVATION DATUM CHANGE	0.00
STATION POSITIONS	
IN GRID COORDS	0
IN LAT/LONG	1

QUE RIVER MINE AREA GRAVITY SURVEY DECEMBER 1980

12/81

ST NUMBER	LATITUDE			LONGITUDE			HEIGHT	OBS GRAV	THEO GRAV	CORRN	BA	COMMENT
8051.6001	41	35	57.4	145	41	47.5	695.40	980.17077	980.32319	1.47	-14.14	
8051.6002	41	35	55.8	145	41	43.6	683.70	980.17242	980.32315	1.48	-14.75	
8051.6003	41	35	57.7	145	41	48.5	698.50	980.17045	980.32320	1.44	-13.89	
8051.6004	41	35	58.0	145	41	49.5	699.20	980.17009	980.32321	1.44	-14.12	
8051.6005	41	35	58.3	145	41	50.4	700.60	980.16969	980.32321	2.00	-13.69	

5

APPENDIX 5

Program KYD2KM

```
  KYD2KM      CONVERT FROM KILOYARD GRID TO METRIC GRID                PAGE 0001

C COORDS.  ARE IN FACT IN YARDS AND METRES.
C N.B.  THE METRIC CORRDINATES ARE LACKING THE LEADING 5
      DOUBLE PRECISION EAST,YNORTH,SNUM,OEAST,ONORTH
      DIMENSION A(20),B(10)
      EQUIVALENCE (A(1),B(1))
      DO 10 I=1,5
      READ(4,100) A
      WRITE(6,100) A
  100 FORMAT(20A4)
   10 CONTINUE
C COPY CONTROL INFORMATION FROM HEAD OF FILE
   15 READ(4,200) SNUM,ONORTH,OEAST,ELEV,GOBS,B,CORR
C NOTE REVERSE ORDER FOR COORDINATES
  200 FORMAT(F9.4,2F10.0,F8.2,F10.2,10A1,F7.2)
      IF (SNUM .LT. -9000.0D00) GOTO 40
C END OF DATA IF NEGATIVE
      EAST=0.914664D00*OEAST-0.012023D00*ONORTH+5.7D-11*OEAST
     .     *ONORTH-8.4D-10*(OEAST**2-ONORTH**2)+60756.0D00
      YNORTH=0.012023D00*OEAST+0.914664D00*ONORTH-1.678D-9*OEAST
     .       *ONORTH-2.85D-11*(OEAST**2-ONORTH**2)+4587325.0D00
      YNORTH=YNORTH-5.0D6
      WRITE(6,200) SNUM,EAST,YNORTH,ELEV,GOBS,B,CORR
      GOTO 15
   40 CONTINUE
      WRITE(6,201)
  201 FORMAT('-999999.')
      ENDFILE 6
      STOP
      END
```

34-AS/1

21/21